§ 3 Résolution d'équations avec Mathematica

§ 3.1 Introduction

Mathematica dispose de plusieurs commandes et méthodes qui peuvent servir à résoudre des équations ou des systèmes d'équations : **Factor**, **NSolve**, **Solve**, **FindRoot**, **FixedPoint**, **LinearSolve, Reduce, ...** Dans ce chapitre, nous utiliserons préférentiellement la méthode **Reduce[...]**, mais nous constaterons que, dans certains cas particuliers, il sera nécessaire de faire appel à d'autres méthodes.

■ Symbole d'égalité

En *Mathematica*, le symbole = désigne une affectation (ou assignation) qui consiste à attribuer une valeur à une variable. Par exemple

```
x = 5
```

```
5
```

Par contre, == désigne le symbole d'égalité dont la valeur est vrai ou faux

```
x == 1
```

```
False
```

```
x == 5
```

```
True
```

C'est le symbole d'égalité qui permet de poser une équation. L'équation comporte habituellement une ou plusieurs inconnue(s), c'est-à-dire une ou plusieurs variable(s) non affectée(s)

```
Clear[x];
```

$$\frac{1}{2}x + \frac{2}{3} == 5$$

$$\frac{2}{3} + \frac{x}{2} == 5$$

§ 3.2 Méthode Reduce

■ Résolution de systèmes d'équations ou d'inéquations avec Reduce[...]

Nous avons éjà étudié, dans le chapitre **Initiation à *Mathematica***, *§ 2 Premiers principes*, comment résoudre des systèmes d'équations ou d'inéquations, en particulier polynomiales comme par exemple

$$\text{N}\left[\text{Reduce}\left[x^3 == 3 - x,\ x,\ \text{Reals}\right]\right]$$

$$x == 1.21341$$

Le système d'équations ou d'inéquations peut être littéral, par exemple

```
Reduce[m x + p == 0,  x,  Reals]
```

$$x == 0. - 0.223591\ p$$

Rappelons que

&& se lit "**et**"; ∧ se lit "**et**";

|| se lit "**ou**"; $\vee$ se lit "**ou**".

Ainsi, la réponse du dernier exemple s'interprète comme une discussion:

1-er cas: si $p = 0$ et $m = 0$, alors tout x est solution

ou

2-ème cas: si $m \neq 0$, alors $x = -\frac{p}{m}$.

Dans le 3-ème cas $m = 0$ et $p \neq 0$ (qui n'apparaît pas formellement), l'équation n'a pas de solution.

Nous apportons ici quelques compléments. **Reduce[...]** est capable de résoudre un système d'équations ou d'inéquations rationnel, par exemple

$$\texttt{Reduce}\left[x \le \frac{1-2x}{x+5},\ x,\ \texttt{Reals}\right]$$

$$x \le \frac{1}{2}\left(-7-\sqrt{53}\right) \ || \ -5 < x \le \frac{1}{2}\left(-7+\sqrt{53}\right)$$

Voici par exemple une inéquation irrationnelle:

$$\texttt{N}\left[\texttt{Reduce}\left[x \ge \frac{\sqrt{1-2x}}{x+2},\ x,\ \texttt{Reals}\right]\right]$$

$$-2.89932 \le x < -2. \ || \ 0.286103 \le x \le 0.5$$

On remarquera que **Reduce** a automatiquement tenu compte de l'ensemble de définition de l'expression.
Il est possible de déterminer l'ensemble de définition d'une fonction irrrationnelle:

$$\texttt{Reduce}\left[\frac{\sqrt{1-2x}}{x+2} < 0 \bigvee \frac{\sqrt{1-2x}}{x+2} \ge 0,\ x,\ \texttt{Reals}\right]$$

$$x < -2 \ || \ -2 < x \le \frac{1}{2}$$

Reduce peut aussi résoudre des équations et inéquations exponentielles ou logarithmiques:

$$\texttt{Reduce}[2^x == 4^x - 13,\ x,\ \texttt{Reals}]$$

$$x == \frac{\texttt{Log}\left[\frac{1}{2}\left(1+\sqrt{53}\right)\right]}{\texttt{Log}[2]}$$

$$\texttt{N}[\texttt{Reduce}[2^x == x + 3,\ x,\ \texttt{Reals}]]$$

$$x == -2.8625 \ || \ x == 2.44491$$

$$\texttt{Reduce}\left[\texttt{Log}\left[\frac{x}{3}\right] \ge \texttt{Log}[1-2x],\ x,\ \texttt{Reals}\right]$$

$$\frac{3}{7} \le x < \frac{1}{2}$$

$$\texttt{Reduce}[\texttt{Log}[x] == x,\ x,\ \texttt{Reals}]$$

`False`

Il est possible de déterminer l'ensemble de définition d'une fonction logarithmique, puis de simplifier le résultat obtenu:

$$\texttt{Reduce}\left[\frac{\texttt{Log}[1-2x]}{x+3} < 0 \bigvee \frac{\texttt{Log}[1-2x]}{x+3} \ge 0,\ x,\ \texttt{Reals}\right]$$

$$x < -3 \ || \ 0 < x < \frac{1}{2} \ || \ -3 < x \le 0$$

$$\text{Simplify}\left[\text{Reduce}\left[\frac{\text{Log}[1-2\,x]}{x+3}<0 \bigvee \frac{\text{Log}[1-2\,x]}{x+3}\geq 0,\ x,\ \text{Reals}\right]\right]$$

$$x<-3\ ||\ -3<x<\frac{1}{2}$$

Dans la solution des systèmes d'équations ou d'inéquations trigonométriques, le symbole ***C*[1]** désigne un entier relatif quelconque

$$\text{Reduce}\left[\text{Cos}[x]==\text{Sin}\left[x-\frac{\pi}{2}\right],\ x,\ \text{Reals}\right]$$

$$C[1]\in \text{Integers}\ \&\&\ \left(x==-\frac{\pi}{2}+2\,\pi\,C[1]\ ||\ x==\frac{\pi}{2}+2\,\pi\,C[1]\right)$$

Reduce[...] nous donne l'ensemble de toutes les solutions, même s'il en existe une infinité. Le résultat précédent se laisse immédiatement traduire dans les notations mathématiques usuelles de la trigonométrie

$$x=-\frac{\pi}{2}+k\,2\,\pi \quad \text{ou} \quad x=\frac{\pi}{2}+k\,2\,\pi \quad \text{où} \quad k\in\mathbb{Z}$$

Reduce[...] résoud aussi les systèmes d'équations ou d'inéquations comportant des valeurs absolues:

```
Reduce[Abs[x] == x, x, Reals]
```

$$x\geq 0$$

Reduce[...] peut aussi rédoudre un système d'équations à plusieurs inconnues. Dans ce cas, le deuxième argument consiste en la liste des inconnues:

$$\text{Reduce}\left[x^2+y-12==0 \bigwedge x+5\,y==9,\ \{x,\ y\},\ \text{Reals}\right]$$

$$\left(x==\frac{1}{10}\left(1-\sqrt{1021}\right)\ ||\ x==\frac{1}{10}\left(1+\sqrt{1021}\right)\right)\ \&\&\ y==12-x^2$$

On trouvera encore des informations supplémentaires dans la partie facultative qui se trouve à la fin de ce § 3.

■ Limites de Reduce[...]

Reduce[...] résoud les systèmes d'équations ou d'inéquations par des transformations algébriques que l'on peut exprimer sous la forme d'équivalences logiques

```
Reduce[Cos[x] == x, x, Reals]

Reduce::nsmet : This system cannot be solved with the methods available to Reduce. >>

Reduce[Cos[x] == x, x, Reals]
```

Un tel message signifie que l'équation doit être résolue par une autre méthode, en l'occurence par une méthode numérique, en particulier par la méthode **FindRoot[...]**.

§ 3.3 Méthode FindRoot

La méthode **FindRoot** est une méthode numérique (consultez l'*Aide*). Nous l'utiliserons essentiellement dans le cas où la méthode **Reduce[...]** échoue.
Soit par exemple à résoudre l'équation $\cos(x) = x$.

```
Reduce[Cos[x] == x, x, Reals]

Reduce::nsmet : This system cannot be solved with the methods available to Reduce. >>

Reduce[Cos[x] == x, x, Reals]
```

Méthode de Newton

Par une méthode graphique, déterminons d'abord une valeur de démarrage x_0.

```
Plot[Cos[x] - x, {x, 0, π/2}]
```




```
x0 = π/4.
```

0.785398

Lorsqu' on donne une seule valeur de démarrage x_0, *Mathematica* utilise la méthode de Newton.

```
r = FindRoot[Cos[x] == x, {x, x0}]
```

{x → 0.739085}

Demandons d'afficher toutes les décimales calculées :

```
NumberForm[r, 16]
```

{x → 0.7390851332151611}

■ Méthode de la sécante

Lorsqu'on donne un encadrement initial [*a*, *b*], *Mathematica* utilise une variante de la méthode de la sécante (voir § 2.3) qui donne un nouvel encadrement de la solution.

```
a = 0; b = π/2;
re = FindRoot[Cos[x] == x, {x, {a, b}}]
```

{x → {0.739085, 0.739085}}

```
{r1, r2} = x /. re
```

{0.739085, 0.739085}

```
r = (r1 + r2)/2
```

0.739085

```
NumberForm[r, 16]
```

```
0.7390851332151607
```

■ Méthode de Newton pour un système

Quoique la question dépasse notre programme d'études, indiquons que **FindRoot** peut être appliqué à un système d'équations.

$$\texttt{FindRoot}\left[\left\{\texttt{Cos}[x] == y - \frac{\pi}{12},\ \texttt{Sin}[y] == \sqrt{x - \frac{1}{4}}\right\},\ \{x, 1\},\ \{y, 1\}\right]$$

```
{x → 0.870025, y → 0.906607}
```

Cette possibilité justifie l'intérêt que nous portons aux méthodes itératives de type point fixe.

§ 3.4 Méthode FixedPoint

Parce que c'est l'objet de notre étude, il peut arriver que nous voulions définir précisément la méthode de type point fixe que nous voulons utiliser. Décidons de résoudre l'équation

$$x^3 - \sqrt{x + 2} = 0$$

au moyen de la méthode pseudo Newton.

$$\texttt{Clear}[f];$$
$$f[x\_] := x^3 - \sqrt{x + 2};$$
$$\texttt{Plot}[f[x], \{x, 0, 2\}]$$



$$a = 1;\ b = 1.5;\ x0 = \frac{a + b}{2};$$
$$m = \frac{f[b] - f[a]}{b - a};$$
$$\texttt{Clear}[g];$$
$$g[x\_] := x - \frac{f[x]}{m}$$

```
? FixedPoint
```

```
FixedPoint[f, expr] commence par expr, puis applique
  f à plusieurs reprises jusqu’à ce que le résultat demeure inchangé.
```

La méthode **FixedPoint[g, x0]** itère la fonction g à partir de x_0 jusqu'à ce que le résultat ne change plus (c'est-à-dire jusqu'à ce que l'on parvienne à un point fixe) :

```
r = FixedPoint[g, x0]
```

```
1.21486
```

```
NumberForm[r, 16]
```

```
1.214862322488425
```

Avec la commande **FixedPointList**, , on peut observer tous les résultats intermédiaires. Compte tenu de la précision atteinte (16 chiffres significatifs), on remarquera la grande vitesse de convergence (seulement 13 itérations).

```
NumberForm[FixedPointList[g, x0], 16]
```

```
{1.25, 1.216383177834951, 1.214970477592186, 1.214870139206551,
 1.214862888070812, 1.214862363414771, 1.214862325449931,
 1.214862322702725, 1.214862322503932, 1.214862322489547, 1.214862322488506,
 1.21486232248843, 1.214862322488425, 1.214862322488425, 1.214862322488425}
```

Compte tenu de la précision atteinte (16 chiffres significatifs), on remarquera la grande vitesse de convergence (seulement 13 itérations).

§ 3.5 Exercices

■ Exercice 3 - P 1

D'un cône, on donne sa hauteur h et son aire totale A.
Calculez son rayon r en résolvant l'équation du problème 1 - P 1 avec *Mathematica*.

Indications

Il s'agit d'un calcul littéral.
Complétez l'équation par une condition afin de n'obtenir qu'une seule réponse.

■ Exercice 3 - P 2

D'un triangle rectangle, on connait son aire A et la projection p d'une cathète sur l'hypoténuse.
Calculez les longueurs des trois côtés a, b, c en résolvant les équations du problème 1 - P 2 avec *Mathematica*.
Données numériques : $A = 975$, $p = 20$.

■ Exercice 3 - P 3

Une sphère homogène de rayon $r = 0.3\,m$, de masse volumique $\rho_1 = 900\,\frac{\text{kg}}{m^3}$ flotte sur un liquide de masse volumique $\rho_0 = 1000\,\frac{\text{kg}}{m^3}$.
Calculez la hauteur h de la partie émergeante en résolvant l'équation du problème 1 - P 3 avec *Mathematica*.

■ Exercice 3 - P 4

Une cuve cylindrique est posée horizontalement sur le sol.
Donnée numérique $r = 50$.
Graduez une jauge, c'est-à-dire dessinez une bande de papier portant les graduations 10 %, 20 %, ... comme l'indique la figure ci-dessous.

```
—— 100 %

—— 90 %
—— 80 %
—— 70 %
—— 60 %
—— 50 %
—— 40 %
—— 30 %
—— 20 %
—— 10 %

—— 0 %
```

a) Première version: résoudre les équations l'une après l'autre, sans utiliser de listes

Indications

Il s'agit d'abord de résoudre 8 équations du problème 1 - P 4 et
de former la liste des angles qui correspondent à chaque graduation:

$\alpha_0 = 0$;
α_1 est la solution de l'équation pour le taux de remplissage t = 0.1;
α_2 est la solution de l'équation pour le taux de remplissage t = 0.2;
α_3 est la solution de l'équation pour le taux de remplissage t = 0.3;
α_4 est la solution de l'équation pour le taux de remplissage t = 0.4;
$\alpha_5 = \pi$;
α_6 est la solution de l'équation pour le taux de remplissage t = 0.6;
α_7 est la solution de l'équation pour le taux de remplissage t = 0.7;
α_8 est la solution de l'équation pour le taux de remplissage t = 0.8;
α_9 est la solution de l'équation pour le taux de remplissage t = 0.9;
$\alpha_{10} = 2\pi$.

Il s'agit ensuite de former la liste des hauteurs qui correspondent à chaque angle

$h_0 = 0$;
h_1 est la hauteur qui correspond à l'angle α_1;
h_2 est la hauteur qui correspond à l'angle α_2;
h_3 est la hauteur qui correspond à l'angle α_3;
h_4 est la hauteur qui correspond à l'angle α_4;
$h_5 = r$;
h_6 est la hauteur qui correspond à l'angle α_6;
h_7 est la hauteur qui correspond à l'angle α_7;
h_8 est la hauteur qui correspond à l'angle α_8;
h_9 est la hauteur qui correspond à l'angle α_9;
$h_{10} = 2r$.

Il faut enfin dessiner une liste de traits horizontaux :

```
Show[Graphics[{
        Line[{{0,h0},{5, h0}}], Line[{{0,h1},{5, h1}}],
        Line[{{0,h2},{5, h2}}], Line[{{0,h3},{5, h3}}],
        Line[{{0,h4},{5, h4}}], Line[{{0,h5},{5, h5}}],
        Line[{{0,h6},{5, h6}}], Line[{{0,h7},{5, h7}}],
        Line[{{0,h8},{5, h8}}], Line[{{0,h9},{5, h9}}],
        Line[{{0,h10},{5, h10}}], Text["0 %",{5,h0},{-1,0}],
        Text["10 %",{5,h1},{-1,0}], Text["20 %",{5,h2},{-1,0}],
        Text["30 %",{5,h3},{-1,0}], Text["40 %",{5,h4},{-1,0}],
        Text["50 %",{5,h5},{-1,0}], Text["60 %",{5,h6},{-1,0}],
        Text["70 %",{5,h7},{-1,0}], Text["80 %",{5,h8},{-1,0}],
        Text["90 %",{5,h9},{-1,0}], Text["100 %",{5,h10},{-1,0}]},
        PlotRange→{{0,20},{-10,110}}, AspectRatio →5]];
```

b) Deuxième version (facultative) : écrire un programme plus compact en utilisant Table[...]

■ Exercice 3 - P 5

Pour rembourser un emprunt de Fr 8'200.--, une banque demande de verser 6 annuités de Fr 2'000.--
Calculez le taux d'intérêt *i* en résolvant l'équation du problème 1 - P 5 avec *Mathematica*.

§ 3.6 Partie facultative

■ Si le lecteur a pris de l'avance, il est invité à étudier maintenant le paragraphe 2.3

■ Résolution de systèmes d'équations ou d'inéquations avec Reduce[...] (Supplément facultatif)

■ Expression des solutions de Reduce sous la forme d'une liste de valeurs (ensemble des solutions)

Lorsqu'on doit enchaîner des calculs, les résultats des calculs précédents doivent souvent être transmis comme données aux calculs qui suivent. Il est souhaitable que cette transmission soit faite par *Mathematica* sans que l'utilisateur doive effectuer des copier/coller.

```
s = Reduce[x^2 == 3 - x, x, Reals]
```

$$x == \frac{1}{2}\left(-1-\sqrt{13}\right) \;||\; x == \frac{1}{2}\left(-1+\sqrt{13}\right)$$

Une première méthode consiste à collecter les réponses sous la forme d'une liste de valeurs que nous nommons ensemble des solutions.

Pour transformer **Or[...]** en **List[...]**:

```
Apply[List, s]
```

$$\left\{x == \frac{1}{2}\left(-1-\sqrt{13}\right),\; x == \frac{1}{2}\left(-1+\sqrt{13}\right)\right\}$$

Pour remplacer chaque symboles d'égalité (== ou **Equal**) par le membre de droite correspondant

```
s /. Equal[x, vx_] → vx
```

$$\frac{1}{2}\left(-1-\sqrt{13}\right) \;||\; \frac{1}{2}\left(-1+\sqrt{13}\right)$$

En combinant les deux opérations, on peut passer à l'ensemble des solutions

```
es = Apply[List, s] /. Equal[x, vx_] → vx
```

$$\left\{\frac{1}{2}\left(-1-\sqrt{13}\right),\ \frac{1}{2}\left(-1+\sqrt{13}\right)\right\}$$

Une autre manière d'arriver à un ensemble des solutions consiste à faire appel aux méthodes **Solve[...]** ou **NSolve[...]** qui expriment les réponses sous la forme de listes de substitutions.

```
NSolve[x^2 == 3 - x, x]
```

```
{{x → -2.30278}, {x → 1.30278}}
```

```
es = x /. NSolve[x^2 == 3 - x, x]
```

```
{-2.30278, 1.30278}
```

```
Solve[x^2 == 3 - x, x]
```

$$\left\{\left\{x\to\frac{1}{2}\left(-1-\sqrt{13}\right)\right\},\ \left\{x\to\frac{1}{2}\left(-1+\sqrt{13}\right)\right\}\right\}$$

```
es = x /. Solve[x^2 == 3 - x, x]
```

$$\left\{\frac{1}{2}\left(-1-\sqrt{13}\right),\ \frac{1}{2}\left(-1+\sqrt{13}\right)\right\}$$

Pour montrer quel usage on peut faire de cet ensemble des solutions, supposons qu'il s'agisse d'une liste d'abscisses de points sur le graphe d'une fonction f donnée

```
Clear[f]; f[x_] := x^2;
```

Pour former les coordonnées des points, on peut utiliser l'un des procédés suivants

```
Table[{es[[j]], f[es[[j]]]}, {j, 1, 2}]
```

$$\left\{\left\{\frac{1}{2}\left(-1-\sqrt{13}\right),\ \frac{1}{4}\left(-1-\sqrt{13}\right)^2\right\},\ \left\{\frac{1}{2}\left(-1+\sqrt{13}\right),\ \frac{1}{4}\left(-1+\sqrt{13}\right)^2\right\}\right\}$$

```
ys = Map[f, es]; Transpose[{es, ys}]
```

$$\left\{\left\{\frac{1}{2}\left(-1-\sqrt{13}\right),\ \frac{1}{4}\left(-1-\sqrt{13}\right)^2\right\},\ \left\{\frac{1}{2}\left(-1+\sqrt{13}\right),\ \frac{1}{4}\left(-1+\sqrt{13}\right)^2\right\}\right\}$$

```
N[Table[{es[[j]], f[es[[j]]]}, {j, 1, 2}]]
```

```
{{-2.30278, 5.30278}, {1.30278, 1.69722}}
```

```
ys = Map[f, es]; N[ Transpose[{es, ys}]]
```

```
{{-2.30278, 5.30278}, {1.30278, 1.69722}}
```

On peut aussi demander d'afficher tous les chiffres significatifs calculés:

```
NumberForm[N[Transpose[{es, ys}]], 16]
```

```
{{-2.302775637731995, 5.302775637731996}, {1.302775637731995, 1.697224362268005}}
```

■ Expression des solutions de Reduce sous la forme d'une liste de substitutions

Lorsqu'on doit enchaîner des calculs, les résultats des calculs précédents doivent souvent être transmis comme données aux calculs qui suivent. Il est souhaitable que cette transmission soit faite par *Mathematica* sans que l'utilisateur doive effectuer des copier/coller.

Une deuxième méthode consiste à collecter les réponses sous la forme d'une liste de substitutions.

```
s = Reduce[x^2 == 3 - x, x, Reals]
```

$x == \frac{1}{2}\left(-1-\sqrt{13}\right) \,||\, x == \frac{1}{2}\left(-1+\sqrt{13}\right)$

Pour transformer **Or[...]** en **List[...]**:

```
Apply[List, s]
```

$\left\{x == \frac{1}{2}\left(-1-\sqrt{13}\right),\ x == \frac{1}{2}\left(-1+\sqrt{13}\right)\right\}$

Pour remplacer chaque symbole d'égalité (== ou **Equal**) par un symbole de substitution (→ ou **Rule**):

```
s /. Equal → Rule
```

$\left(x \to \frac{1}{2}\left(-1-\sqrt{13}\right)\right) \,||\, \left(x \to \frac{1}{2}\left(-1+\sqrt{13}\right)\right)$

En combinant les deux opérations, on peut passer à l'ensemble des substitutions

```
ls = Apply[List, s] /. Equal → Rule
```

$\left\{x \to \frac{1}{2}\left(-1-\sqrt{13}\right),\ x \to \frac{1}{2}\left(-1+\sqrt{13}\right)\right\}$

Une autre manière d'arriver à une liste de substitutions consiste à faire appel aux méthodes **Solve[...]** ou **NSolve[...]** qui expriment directement les réponses sous cette forme:

```
ls = NSolve[x^2 == 3 - x, x]
```

{{x → -2.30278}, {x → 1.30278}}

```
ls = Solve[x^2 == 3 - x, x]
```

$\left\{\left\{x \to \frac{1}{2}\left(-1-\sqrt{13}\right)\right\},\ \left\{x \to \frac{1}{2}\left(-1+\sqrt{13}\right)\right\}\right\}$

Pour montrer quel usage on peut faire de cette liste de substitutions, supposons qu'il s'agisse d'une liste d'abscisses de points sur le graphe d'une fonction *f* donnée

```
Clear[f]; f[x_] := x^2;
```

Pour former les coordonnées des points, on peut utiliser le procédé suivant

```
{x, f[x]} /. ls
```

$\left\{\left\{\frac{1}{2}\left(-1-\sqrt{13}\right),\ \frac{1}{4}\left(-1-\sqrt{13}\right)^2\right\},\ \left\{\frac{1}{2}\left(-1+\sqrt{13}\right),\ \frac{1}{4}\left(-1+\sqrt{13}\right)^2\right\}\right\}$

```
{x, f[x]} /. N[ls]
```

{{-2.30278, 5.30278}, {1.30278, 1.69722}}

On peut aussi demander d'afficher tous les chiffres significatifs calculés:

```
NumberForm[N[Transpose[{ls, ys}]], 16]
```

{{{x → -2.302775637731995}, 5.302775637731996},
 {{x → 1.302775637731995}, 1.697224362268005}}